Tarski's Geometry and the Euclidean Plane in Mizar

Adam Grabowski
Institute of Informatics
University of Białystok
ul. Ciołkowskiego 1 M
15-245 Białystok, Poland
adam@math.uwb.edu.pl

Roland Coghetto
Rue de la Brasserie, 5
B-7100 La Louviére
Belgium
roland_coghetto@hotmail.com

Abstract

We discuss the formal approach to Tarski geometry axioms modelled with the help of the Mizar computerized proof assistant system. We try however to go much further from the use of simple predicates in the direction of the use of structures with their inheritance, attributes as a tool of more human-friendly namespaces for axioms, and registrations of clusters to obtain more automation (with the possible use of external equational theorem provers like Otter/Prover9). The formal proof that Euclidean plane satisfies all eleven axioms proposed by Tarski is an essential development, allowing to show the independence of the parallel postulate, one of the items from "Top 100 mathematical theorems".

1 Introduction

For years, foundations of geometry attracted a lot of interest of resarchers from various areas of mathematics. From the very beginnings, human thought was stimulated by geometrical objects, however from the modern viewpoint of automated theorem-provers, diagrams can deliver some really tough problems. Here an important example is the possibility of ruler-and-compass construction: impossibility of trisecting the angle and doubling the cube (as two out of four problems of antiquity), where the treatment of constructible numbers is way more efficient from the formal point of view.

The choice of the topic is not accidental – recent code available in Coq [BN12] and the use of automated equational provers caught an eye of researchers and, as a by-product, some results which shed some new light on the axiomatization of geometry were published. One of the bright milestones was also the publication of the new issue of the classical textbook *Metamathematische Methoden in der Geometrie* by Schwabhäuser, Szmielew, and Tarski (SST) with the foreword of Beeson.

2 Mizaring Affine Geometry

In Tarski's system of axioms [TG99] the only primitive geometrical notions are points, the ternary relation B of "soft betweenness" and quaternary relation $\equiv$ of "equidistance" or "congruence of segments". The axioms are reflexivity, transitivity, and identity axioms for equidistance; the axiom of segment construction; reflexivity, symmetry, inner and outer transitivity axioms for betweenness; the axiom of continuity, and some others. The original set consisted of 20 axioms for two-dimensional Euclidean geometry and was constructed in 1926–27, submitted for publication in 1940, and finally appeared in 1967 in a limited number of copies. There are many modifications of this system, and Gupta's work in this area [Gup65] offers an important simplification.

Another notable axiomatization, proposed by Hilbert [Hil80] in 1899, has three sorts: planes, points, and lines, and three relations: betweenness, containment, and congruence. In this sense it is a little bit more complex than Tarski's (but not necessarily in terms of numbers of axioms as it has also 20 of these). The two approaches establish a geometrical framework, in which theorems can be proven logically (remember Euclid's *Elements* proofs are mainly pictures or graphical constructions and rely heavily on intuition). This allows to use computerized theorem prover in order to find the proof or proof checker to check the theory for its correctness. We deal with the proof checker Mizar based on classical first order logic and Tarski-Grothendieck set theory (a variant of Zermelo-Fraenkel) and describe, how Tarski's theory was built formally.

Geometry in the Mizar Mathematical Library is based on eight various structures, so it raises communication issues between various approaches. The last article was `PROJPL_1` dated back to 1994, and the series was not very actively developed (with the exception of the paper of the combinatorial Grassmanians `COMBGRAS`). But in 1990 Mizar articles on geometry were about 30% of the whole Mizar repository (out of 140 files). Of course, after that time revisions of this specific area were quite active: one of the main streams (done also by the current author) was to get separate axioms of selected properties instead of a large block for *the mode* (i.e. the constructor of the type in Mizar). Affine approach to geometry was less important as there was another big challenge which for fifteen years stimulated geometry (in analytical setting, however): the proof of the Jordan Curve Theorem.

As notable affine geometrical facts already formalized in Mizar we can enumerate:

- Hessenberg's theorem – `HESSENBE`;
- Desargues theorem (present in "Top 100 mathematical theorems") – `ANPROJ_2`;
- Pasch configuration axioms – `PASCH`;
- Fano projective spaces – `PROJRED1`;
- Desarguesian projective planes – `PROJRED2`;
- Pappus, Minor, Major and Trapezium Desargues axioms – `AFF_2`;
- Minkowskian geometry – `ANALORT`.

In our opinion, the unifiying approach in Tarski's spirit could be quite useful to bind all of these geometrical results together. Among another significant facts in geometry we can point out Morley trisector theorem, Ceva, and Menelaus theorem. These facts however deal with Euclidean plane, so the proofs are in the area of analytic geometry; they were developed more than ten years later than the foundations of geometry in Mizar, when Euclidean spaces were more thoroughly explored in MML.

The distinction between classical and abstract mathematics (i.e. the one based on ordinary axioms of set theory, and all those using the notion of a structure, respectively) is important from the viewpoint of the organization of the Mizar repository. We had to choose between two paths:

- it is possible to formulate Tarski's axioms without the use of a structure, and also set theory could be meaningless for that framework, only the classical logic with Mizar predicates is enough;

- the use of Mizar structures forces us paradoxically to use fundamentals of set theory – defining a signature of Tarski's plane needed to give a type of congruence of segments and betweenness relation, which was set-theoretic (Mizar language is typed, and it the earlier case one should also give a type at least to points, but it can be defined as Mizar `object`).

The latter was also chosen by us as the whole geometry in MML is written in abstract style (as the majority of MML) as structures in Mizar are present for a long time. Even if in ordinary mathematical tradition they are considered as ordered tuples, in the implementation in Mizar they are treated rather as partial functions, with selectors as arguments, and ordinary inheritance mechanism (with polymorphic enabled, which will be extensively used in our formalizations).

```
definition
  struct (1-sorted) TarskiPlane
    (# carrier -> set,
    Betweenness -> Relation of [:the carrier, the carrier:], the carrier,
    Equidistance -> Relation of [:the carrier, the carrier:], [:the carrier, the carrier:] #);
end;
```

The betweenness relation can be treated as a ternary relation, but the choice of this concrete model was quite arbitrary as the difference between dealing with ternary relations and relations between ordered pairs and elements will not cause any major problems later (we use mainly predicates). Then we can use a type `POINT of S` just for elements of structures S, and predicates `between a,b,c` and `a,b equiv c,d` as `[[a,b],c] in the Betweenness of S` and `[[a,b],[c,d]] in the Equidistance of S`, respectively.

3 The First Part of Tarski's Axiomatics

Original version of Mizar formalization of Tarski's axioms done by William Richter with the help of `miz3` did not contained any existence proofs. This essentially caused the lack of the appropriate Mizar type. By using attributes instead of predicates we can have modular building of a complex structure, all other can be reused; hence in our Mizar article we focus on pure betweenness-equidistance part, not really mentioning the question of dimensions. We proved 44 theorems (properties of the predicates), with the Gupta's proof of Hilbert's I1 axiom (that two distinct points determine a line).

Our Mizar versions of Tarski's axioms have descriptive names, and follow the ones from SST (using $\equiv$ for congruence of segments and B for betweenness relation):

- `CongruenceSymmetry` (A1):
  $\forall_{a,b}\ ab \equiv ba$,
- `CongruenceEquivalenceRelation` (A2):
  $\forall_{a,b,p,q,r,s}\ ab \equiv pq \wedge ab \equiv rs \Rightarrow pq \equiv rs$,
- `CongruenceIdentity` (A3):
  $\forall_{a,b,c}\ ab \equiv cc \Rightarrow a = b$,
- `SegmentConstruction` (A4):
  $\forall_{a,q,b,c} \exists_x\ B(q,a,x) \wedge ax \equiv bc$,
- `BetweennessIdentity` (A6):
  $\forall_{a,b}\ B(a,b,a) \Rightarrow a = b$,
- and `Pasch` (A7):
  $\forall_{a,b,p,q,z}\ B(a,p,z) \wedge B(p,q,z) \Rightarrow \exists_x\ B(p,x,b) \wedge B(q,x,a)$.

One attribute has the form which substantially differs from SST version: in order to shorten the notation, we introduced technical predicate

```
definition
  let S be TarskiPlane;
  let a, b, c, x, y, z be POINT of S;
  pred a,b,c cong x,y,z means          :: GTARSKI1:def 3
    a,b equiv x,y & a,c equiv x,z & b,c equiv y,z;
end;
```

denoting essentially SSS predicate for triangles. Using this notion, SST axiom (A5) could be encoded as follows:

```
definition let S be TarskiPlane;
  attr S is satisfying_SAS means       :: GTARSKI1:def 9
     for a, b, c, x, a1, b1, c1, x1 being POINT of S holds
        a <> b & a,b,c cong a1,b1,c1 &
        between a,b,x & between a1,b1,x1 &
        b,x equiv b1,x1 implies c,x equiv c1,x1;
end;
```

that is,

$$\forall_{a,b,c,x,a',b',c',x'}\ (a \neq b\ \wedge\ ab \equiv a'b'\ \wedge\ bc \equiv b'c'\ \wedge\ ac \equiv a'c'\ \wedge \\ \wedge\ B(a,b,x)\ \wedge\ B(a',b',x')\ \wedge\ bx \equiv b'x')\ \Rightarrow\ cx \equiv c'x'.$$

Having separate attributes for distinct axioms could had already shown its usefulness in various geemetrical settings. It could also allow later for defining equivalent axiom systems for Tarski geometry (and due to mechanism of clusters this equivalence will be obvious for the checker, once proven). But also additional attribute, `satisfying_Tarski-model`, was defined as a shorthand for the above seven axioms.

4 Adding Metric Ingredient

It is well known fact that every metric space can be equipped with the natural topology. This informally obvious mathematical property brings some unexpected difficulties when dealing with structures if automated proof-assistants play a role. Namely then, if one considers topological spaces in a quite natural way, that is $\langle U, \tau \rangle$, and metric space as $\langle U, d \rangle$, respectively, one can rather naturally merge both structures into common $\langle U, \tau, d \rangle$.

During the formalization of the Jordan Curve Theorem in Mizar, however, another approach was chosen (and pushed consequently until the successful finale): `Euclid 2` denoted metric space concerned with the Euclidean plane, and then special functor converting any metric space into the topoogical space was applied to obtain `TOP-REAL 2` (of course the conversion can be made for arbitrary natural number n, not necessarily 2, but Jordan curves deal with two-dimensional case). The basic signature for metric spaces are `MetrStruct`, where distance is a function defined on the Cartesian square of the carrier with the real values. Then, metrics (or pseudo-, quasi-, semimetrics, etc.) can be defined in terms of attributes [KLS90], that is properties of the distance function.

```
definition
  struct (MetrStruct,TarskiPlane) MetrTarskiStr
   (# carrier -> set,
     distance -> Function of [:the carrier, the carrier:], REAL,
     Betweenness -> Relation of [:the carrier, the carrier:], the carrier,
     Equidistance -> Relation of [:the carrier, the carrier:], [:the carrier, the carrier:] #);
end;
```

Then we have two worlds merged: affine, where we have two Tarski's relations, and Euclidean, where in terms of distance function, we can have betweenness relation and the the measure for segments. We argue that, regardless of all the complications caused by merging structures (which increases the number of selectors, hence the chain of notions gets more complicated), such approach – not converting between two contexts, but rather to make reasoning in the world which is successor of both – allows for more flexible reuse of knowledge from two original areas.

```
definition let M be MetrTarskiStr;
  attr M is naturally_generated means       :: GTARSKI1:def 15
    (for a, b, c being POINT of M holds between a,b,c iff b is_Between a,c) &
    (for a, b, c, d being POINT of M holds a,b equiv c,d iff dist (a,b) = dist (c,d));
end;
```

In merged structure, we want to have two segments congruent, if they are equal in terms of distance, and betweenness relation uses the predicate `is_Between`, also given in terms of the sum of distances.

Recalling the previous discussion on the structure merging, the construction of such a space from scratch (essentially more or less modified copy-and-paste work on the proof from [Try90]) can be avoided with the help of some useful tricks, investigating knowledge already present in MML with its possible reuse. For example, based on the geometry on the real line, appropriate geometrical structure from metric structure can be just an extension with properly defined distance.

```
definition
  func TarskiSpace -> MetrTarskiStr equals   :: GTARSKI1:def 22
    the naturally_generated TarskiExtension of RealSpace;
  coherence;
end;
```

Then, we can show that in such extension the metric is well defined, i.e. this space is reflexive, symmetric, and discerning. Furthermore, if we take into account "geometrical part", all newly introduced Tarski's axioms are true.

5 Euclidean Plane Satisfies Tarski's Axioms

At the beginning, we had to construct an example of the structure which satisfies all these axioms.

```
definition let n be Nat;
  func TarskiEuclidSpace n -> MetrTarskiStr equals        :: GTARSKI2:def 1
    the naturally_generated TarskiExtension of Euclid n;
end;
```

Table 1: The statistics of our formalizations

Items	Numbers in [RGA14]	Numbers in [CG16]
attributes	10	4
lines	1522	1926
kBytes	50	73
theorems	47	50

and `TarskiEuclid2Space` is just the shorthand for the above functor with $n = 2$.

First part of the work, which was rather easy, was to show that such Euclidean plane satisfies seven axioms from [RGA14]. It was expressed in terms of cluster registration:

```
registration
  cluster TarskiEuclid2Space -> satisfying_Tarski-model;
  coherence;
end;
```

Then, we defined the remaining four axioms, with both descriptive names and those corresponding to SST namespace, as follows:

```
notation let S be TarskiPlane;
  synonym S is satisfying_Lower_Dimension_Axiom for S is satisfying_A8;
  synonym S is satisfying_Upper_Dimension_Axiom for S is satisfying_A9;
  synonym S is satisfying_Euclid_Axiom for S is satisfying_A10;
  synonym S is satisfying_Continuity_Axiom for S is satisfying_A11;
end;
```

Of course, their meaning is just as in SST. Then, finally we can deal with the remaining axioms:

```
registration
  cluster TarskiEuclid2Space -> satisfying_Lower_Dimension_Axiom satisfying_Upper_Dimension_Axiom
    satisfying_Euclid_Axiom satisfying_Continuity_Axiom;
end;
```

Some of the proofs were proven based on the space with the dimension 2 (mainly those done by the second author), in few places we used Isabelle development and gave our Mizar proofs accordingly. The summary of formalizations in two Mizar files can be found in Table 2.

6 Conclusions

Although the history of the development of the axiom system for geometry by Tarski is not very clear from the beginnings (as the early works by Tarski seem to be postponed by the World War II), and even if the ultimate source of information is

The book by Schwabhäuser, Szmielew and Tarski, reissued with the foreword of Michael Beeson, attracted recently a lot of focus from automated deduction systems. The remarkable item here is of course Narboux's formal development of Tarski's system with the use of Coq [Nar07]. We are planning to include some interesting results from GeoCoq in the Mizar system; Nakasho's et al. MML symbol reference system[1] offers quite user-friendly interface for browsing appropriate content. Of course, GeoCoq provides ready-to-use list of notions, which are connected only with the Tarski's system – that makes the browsing more convenient. Of course, after rescaling all definitions and theorems can be browsed within MML providing a look in the style of GeoCoq project.

We have created in [RGA14] (and `GTARSKI2` [CG16]) complete formal axiomatization of Tarski's geometry which, at least in our opinion, has the advantage of higher readability for ordinary mathematicians than e.g. Coq or Prover9 proof objects. In the same it is tightly connected with another axiomatization of Euclidean plane, due to Hilbert, already available in MML. Important part of our development was the proof that the real Euclidean plane satisfies all Tarski's axioms.

[1]The system can be browsed at `http://webmizar.cs.shinshu-u.ac.jp/mmlfe/current/` with the official version of the Mizar system.

References

[BBG15] Bancerek G., Byliński Cz., Grabowski A., Korniłowicz A., Matuszewski R., Naumowicz A., Pąk K., Urban J.: Mizar: state-of-the-art and beyond, Intelligent Computer Mathematics, *Lecture Notes in Computer Science* 9150, pp. 261–279 (2015)

[Bee16] Beeson M.: Mixing computations and proofs, *Journal of Formalized Reasoning,* 9(1), pp. 71–99 (2016)

[BW14] Beeson M., Wos L.: OTTER proofs in Tarskian geometry, in Proceedings of IJCAR 2014, *Lecture Notes in Computer Science,* 8562, pp. 495–510 (2014)

[Bor60] Borsuk K., Szmielew W.: *Foundations of Geometry,* North-Holland (1960)

[BN12] Braun G., Narboux J.: From Tarski to Hilbert, in *Automated Deduction in Geometry,* T. Ida and J. Fleuriot (eds.) Proceedings of ADG 2012 (2012)

[CG16] Coghetto R., Grabowski A.: Tarski geometry axioms – part II, *Formalized Mathematics*, 24(2), available at `http://mizar.org/library/tarski/` (2016)

[Gra14] Grabowski A.: Efficient rough set theory merging, *Fundamenta Informaticae,* 135(4), pp. 371–385 (2014)

[Gra15] Grabowski A.: Mechanizing complemented lattices within Mizar type system, *Journal of Automated Reasoning,* 55(3), pp. 211–221 (2015)

[GKN10] Grabowski A., Korniłowicz A., Naumowicz A.: Mizar in a nutshell, *Journal of Formalized Reasoning,* 3(2), 153–245 (2010)

[Gup65] Gupta H.N.: Contributions to the axiomatic foundations of geometry, PhD thesis, University of California, Berkeley (1965)

[Hil80] Hilbert D.: *The Foundations of Geometry,* Chicago: Open Court, 2nd ed. (1980)

[KLS90] Kanas, S., Lecko, A., Startek M.: Metric spaces, *Formalized Mathematics,* 1(3), pp. 607–610 (1990)

[Kor15] Korniłowicz A.: Definitional expansions in Mizar, *Journal of Automated Reasoning,* 55(3), pp. 257–268 (2015)

[Mak12] Makarios T.: A mechanical verification of the independence of Tarski's Euclidean Axiom, MSc thesis (2012)

[MF03] Meikle L., Fleuriot J.: Formalizing Hilbert's Grundlagen in Isabelle/Isar, in Proceedings of TPHOLs'03, *Lecture Notes in Computer Science,* 2758, pp. 319–334 (2003)

[Nar07] Narboux J.: Mechanical theorem proving in Tarski's geometry, in *Automated Deduction in Geometry,* F. Botana and T. Recio (eds.), Lecture Notes in Computer Science, 4869, pp. 139–156 (2007)

[NK09] Naumowicz A., Korniłowicz A.: A brief overview of Mizar, in *Theorem Proving in Higher Order Logics 2009,* S. Berghofer, T. Nipkow, Ch. Urban, M. Wenzel (Eds.), Lecture Notes in Computer Science, 5674, 67–72 (2009)

[RGA14] Richter W., Grabowski A., Alama J.: Tarski geometry axioms, *Formalized Mathematics,* 22(2), pp. 167–176 (2014)

[SST83] Schwabhäuser W., Szmielew W., Tarski A.: *Metamathematische Methoden in der Geometrie,* Springer-Verlag (1983)

[Tar59] Tarski A.: What is elementary geometry?, in *Studies in Logic and the Foundations of Mathematics,* North-Holland, pp. 16–29 (1959)

[TG99] Tarski A., Givant S.: Tarski's system of geometry, *The Bulletin of Symbolic Logic,* 5(2), pp. 175–214 (1999)

[Try90] Trybulec, W.A.: Axioms of incidence, *Formalized Mathematics*, 1(1), pp. 205–213 (1990)

[Wie12] Wiedijk F.: A synthesis of the procedural and declarative styles of interactive theorem proving, *Logical Methods in Computer Science,* 8(1):30 (2012)